The Mathematica® Journal

Tzitzeica Curves and Surfaces

Alfonso F. Agnew
Alexandru Bobe
Wladimir G. Boskoff
Bogdan D. Suceava

Tzitzeica curves and surfaces represent early examples of affine-invariant geometrical objects. At the time Gheorghe Tzitzeica was studying these objects, affine differential geometry (ADG) was in its infancy. ADG was motivated by Felix Klein's influential Erlangen program, where a geometry was defined by its set of invariants under a group of symmetries. We find that the issue lends itself well to a relatively elementary discussion suitable for upper-division undergraduates and nonspecialists, while still providing the basic thrust of this elegant subject. Moreover, the topic is an excellent one to illustrate the utility of *Mathematica*'s symbolic manipulation and graphics capabilities. For this reason, the article nicely complements the existing literature on the uses of software in differential geometry (such as [1]), and it provides material that would be useful for inclusion in a differential geometry course either as an application or a project.

1. Introduction

In the early decades following the 1872 publication of Klein's Erlangen program [2], Gheorghe Tzitzeica was studying particular affine invariants in three-dimensional space, leading to what are now known as Tzitzeica curves and Tzitzeica surfaces [3, 4]. The results were among the earliest contributions to the then-budding area of mathematics that now goes by the name affine differential geometry [5, 6]. Tzitzeica curves and surfaces satisfy at each point the metric relation

$$\Gamma = a\, d^{\beta}, \tag{1}$$

where

- For curves, Γ denotes the torsion, d is the distance from the origin to the osculating plane, $\beta = 2$, and $a \in \mathbb{R}$.

- For surfaces, Γ denotes the Gaussian curvature, d is the distance from the origin to the tangent plane, $\beta = 4$, and $a \in \mathbb{R}$.

In our treatment of this topic we aim to provide interesting material with historical remarks suitable for inclusion in an undergraduate differential geometry course, or as an introduction for the nonspecialist. We emphasize how the topic makes particularly good use of *Mathematica,* in the spirit of Alfred Gray's excellent book [1]. We note that such a discussion is absent from the literature so far.

In the second section, we review the basic ideas needed for a discussion of Tzitzeica surfaces: affine invariance and the basic tools of the differential geometry of curves and surfaces in three dimensions. In Section 3 we define Tzitzeica curves and surfaces, provide examples, and explicitly exhibit the affine invariance. This issue is continued in an appendix, where the program is conveniently summarized to allow the user to test whether or not a surface is Tzitzeica by inputting the coordinate functions of a surface. In Section 4 we take up the issue of asymptotic curves and their special relation to Tzitzeica surfaces, eventually proving that on a Tzitzeica surface, asymptotic curves are Tzitzeica curves. Lastly, we make some concluding remarks in the final section.

2. Background

An affine transformation may always be written $x \longrightarrow A\,x + b$, where A is a nonsingular matrix and b represents a translation. If $b = 0$, then we have a *centro-affine* transformation, while if $\det A = 1$, we have an *equi-affine* transformation. In the case of affine geometry, one is concerned with objects or propositions that retain their character when an affine transformation is applied to the underlying space. For example, a circle in the plane is not an object of affine geometry, because under a general affine transformation of the plane, a circle is mapped to an ellipse with nontrivial eccentricity. On the other hand, one can show that an ellipse will always map to an ellipse under an affine transformation, and so the property of "being an ellipse" is invariant under an affine transformation: an ellipse is an object of affine geometry.

Example 1. Here is the unit circle $x^2 + y^2 = 1$.

```
ParametricPlot[{Cos[t], Sin[t]}, {t, 0, 2 Pi}, AspectRatio → 1]
```



We apply the shear transformation $A = \begin{pmatrix} 1 & 1 \\ 0 & 1 \end{pmatrix}$ to the circle to obtain an ellipse.

```
A = (1  1
     0  1);
A.{x, y}
```

```
{x + y, y}
```

```
ParametricPlot[A.{Cos[t], Sin[t]}, {t, 0, 2 Pi},
 AspectRatio → 1]
```



We now recall, in brief, some basic ingredients of the differential geometry of curves and surfaces in three-dimensional space [7]. Consider any *regular curve* $\alpha : I \longrightarrow \mathbb{R}^3$ (taken, without loss of generality, to be of unit speed $\|\alpha'\| = 1$), where I is an interval in $\mathbb{R}$. Associated with α at each of its points is its *orthonormal Frenet frame* $\{\mathbf{T}, \mathbf{N}, \mathbf{B}\}$. Here, $\mathbf{T} = \alpha'$ is the unit tangent vector field, $\mathbf{N} = \mathbf{T}' / \|\mathbf{T}'\|$ is the principal normal vector field, and $\mathbf{B} = \mathbf{T} \times \mathbf{N}$ is the binormal vector field. We note that the *osculating plane* of α at a point is the plane spanned by $\mathbf{T}$ and $\mathbf{N}$; in particular, $\mathbf{B}$ is normal to the osculating plane. We allow a standard abuse of notation and denote the position vector for a point $p = \alpha(t_0)$ on the curve by α. Then, the orthogonal distance from the origin to the osculating plane is the projection $d_{\text{osc}} = \mathbf{B} \cdot \alpha$.

A subset $M \subset \mathbb{R}^3$ is a *surface*, provided that each point of M is contained in a neighborhood that is, in turn, contained in the image of an injective regular map of an open subset of $\mathbb{R}^2$ into M. Furthermore, this injective map (commonly called a *patch*) will be assumed to have a continuous inverse. When working within the image of a single patch in M, as we always will, it is possible to define an unambiguous unit normal vector field $\boldsymbol{U}$. We may then define the *shape operator* S of M at a point p as

$$S_p(v) = -\nabla_v \boldsymbol{U}, \tag{2}$$

where v is any tangent vector at p and $\nabla_v \boldsymbol{U}$ is the *covariant derivative* of $\boldsymbol{U}$ in the direction of v (i.e., the vector field whose Cartesian components result from applying the directional derivative in the direction v to each of the component functions of $\boldsymbol{U}$). Geometrically, the shape operator describes how the surface is bending in the direction v by giving a measure of how the direction of the unit normal (hence the tangent spaces) is changing. The shape operator is, at each point p, a linear operator on the tangent space at p, and the determinant of this operator is the *Gaussian curvature* $K(p)$ of M at p.

Given a patch of M, say $x = x(u, v)$, we may compute $K = \frac{l\,n - m^2}{E\,G - F^2}$, where

$$\begin{aligned} &E = x_u \cdot x_u,\, F = x_u \cdot x_v,\, G = x_u \cdot x_u, \\ &l = x_{uu} \cdot U,\, m = x_{uv} \cdot U,\, n = x_{vv} \cdot U, \end{aligned} \tag{3}$$

and where $x_u = \frac{\partial}{\partial u} x$, $x_{uv} = \frac{\partial^2}{\partial u\, \partial v} x$, etc. We note that x_u and x_v are tangent to M. Furthermore, we note that for a fixed point $p = x\,(u_0, v_0)$ of M, the orthogonal distance d from the origin in $\mathbb{R}^3$ to the tangent plane to M at p is given by the projection of the position vector $x = x\,(u_0, v_0)$ for p onto U. In short,

$$d_{\tan} = \boldsymbol{U} \cdot x. \tag{4}$$

3. Tzitzeica Curves and Surfaces

3.1 Basic Concepts

A curve α in $\mathbb{R}^3$ is called a *Tzitzeica curve* provided there exists a constant $a \in \mathbb{R}$ such that for all points on a,

$$\tau = a\, d_{\text{osc}}{}^2, \tag{5}$$

where τ is the *torsion* of α, defined by $\mathbf{B}' = -\tau\, N$. The torsion provides a measure of the extent to which α is moving out of its osculating plane. Thus, in rough terms, a Tzitzeica curve is a curve for which the motion of α in the direction of $\mathbf{B}$ is in fixed proportion to the square of the $\mathbf{B}$ component of the curve α.

A surface M in $\mathbb{R}^3$ is called a *Tzitzeica surface*, provided there exists a constant $a \in \mathbb{R}$ such that for all points on M,

$$K = a\, d_{\tan}{}^4, \tag{6}$$

where K is the Gaussian curvature of the surface defined above. K provides a measure of the extent to which M, near a point p, is curving like a paraboloid ($K > 0$), hyperboloid ($K < 0$), or a cylinder or plane ($K = 0$). Thus, again in rough terms, a Tzitzeica surface is a surface for which its bending is in fixed proportion to the normal component of the position vector x.

□ 3.2 Example of the Tzitzeica Surface $z = 1/x\,y$

We work in the patch $r(x, y) = (x, y, 1/(x\,y))$ by first defining z. In the following, we include the output explicitly only for the most relevant quantities.

```
z = 1 / (x y);
```

We plot a portion of the surface for a visual reference.

```
ParametricPlot3D[{{x, y, z}, {-x, -y, z}, {-x, y, -z},
  {x, -y, -z}}, {x, 0.1, 4}, {y, 0.1, 4}]
```



We define the patch.

```
r = {x, y, z};
```

We compute the tangent vector fields.

```
rx = D[r, x]; ry = D[r, y];
```

The components of the first fundamental form, E, F, G, may now be computed.

```
e = rx.rx
```

$$1+\frac{1}{x^4\,y^2}$$

```
f = rx.ry
```

$$\frac{1}{x^3\,y^3}$$

```
g = ry.ry
```

$$1+\frac{1}{x^2\,y^4}$$

We define a unit normal vector field using the cross product.

```
u = Cross[rx, ry] / Sqrt[Cross[rx, ry].Cross[rx, ry]] // Simplify
```

$$\left\{\frac{1}{x^2\sqrt{1+\frac{1}{x^2\,y^4}+\frac{1}{x^4\,y^2}}\;y},\ \frac{1}{x\sqrt{1+\frac{1}{x^2\,y^4}+\frac{1}{x^4\,y^2}}\;y^2},\ \frac{1}{\sqrt{1+\frac{1}{x^2\,y^4}+\frac{1}{x^4\,y^2}}}\right\}$$

Once we compute the second derivatives of `r`, we will be in a position to compute the components of the shape operator.

```
rxx = D[rx, x]; rxy = D[rx, y]; ryy = D[ry, y];
```

```
l = u.rxx // Simplify
```

$$\frac{2}{x^3\sqrt{1+\frac{1}{x^2\,y^4}+\frac{1}{x^4\,y^2}}\;y}$$

```
m = u.rxy // Simplify
```

$$\frac{1}{x^2\sqrt{1+\frac{1}{x^2\,y^4}+\frac{1}{x^4\,y^2}}\;y^2}$$

```
n = u.ryy // Simplify
```

$$\frac{2}{x\sqrt{1+\frac{1}{x^2 y^4}+\frac{1}{x^4 y^2}}\ y^3}$$

We may now compute the Gaussian curvature of $z = 1/(x\,y)$.

```
k = (l n - m^2) / (e g - f^2) // Simplify
```

$$\frac{3\,x^4\,y^4}{\left(x^2+y^2+x^4\,y^4\right)^2}$$

The distance to the tangent plane is as follows.

```
d = u.r // Simplify // Expand
```

$$\frac{3}{x\sqrt{1+\frac{1}{x^2 y^4}+\frac{1}{x^4 y^2}}\ y}$$

Compare the following to the curvature above.

```
d^4 // Simplify
```

$$\frac{81\,x^4\,y^4}{\left(x^2+y^2+x^4\,y^4\right)^2}$$

Lastly, we compute the ratio of the Gaussian curvature to the fourth power of the distance to the tangent plane. We find that $z = 1/(x\,y)$ is a Tzitzeica surface, since this ratio is constant over the surface.

```
k / d^4 // Simplify
```

$$\frac{1}{27}$$

□ 3.3 Applying a General Centro-Affine Transformation

We now compute the effect of a general centro-affine transformation on our Tzitzeica surface. We will find that the result will again be a Tzitzeica surface with the same numerical factor $a = 1/27$, up to a consistent volume term. The volume term indicates that, strictly speaking, the Tzitzeica property is invariant under the subgroup of *central equi-affine* (i.e., volume-preserving with no translation) transformations, where the determinant of the representing matrix is 1.

It is impressive to note how difficult and tedious such a calculation (albeit an important one) would be to do by hand, whereas the entire computation takes approximately 15 seconds on a dual 2.5GHz Apple Macintosh G5. Also, we note that by inputting values into the general matrix A, one can determine the effect of a particular centro-affine transformation, and the resulting invariance is exhibited. Of course, the matrix must still be nonsingular.

The computational details are largely the same as above; however, we first apply the general transformation by multiplying the position vector $\boldsymbol{r}$ by the matrix $A = \{\{a_{11}, a_{12}, a_{13}\}, \{a_{21}, a_{22}, a_{23}\}, \{a_{31}, a_{32}, a_{33}\}\}$.

```
z = 1 / (x y);
```

$$\frac{1}{x\,y}$$

```
r = {{a11, a12, a13}, {a21, a22, a23}, {a31, a32, a33}}.{x, y, z}
```

$$\left\{a11\,x + \frac{a13}{x\,y} + a12\,y,\ a21\,x + \frac{a23}{x\,y} + a22\,y,\ a31\,x + \frac{a33}{x\,y} + a32\,y\right\}$$

Compute the tangent vectors and their inner products.

```
rx = D[r, x]; ry = D[r, y]; e = rx.rx; f = rx.ry;
g = ry.ry;
```

The computation of the unit normal vector U is fairly complicated and is often the most CPU intensive (aside from simplification routines, of course).

```
u = Cross[rx, ry] / Sqrt[Cross[rx, ry].Cross[rx, ry]] //
   Simplify;
```

In order to compute the Gaussian curvature, we once again compute second derivatives and their inner products with the unit normal.

```
rxx = D[rx, x]; rxy = D[rx, y]; ryy = D[ry, y];
l = u.rxx // Simplify; m = u.rxy // Simplify;
n = u.ryy // Simplify;
```

We explicitly compute the curvature.

```
k = (l n - m^2) / (e g - f^2) // Simplify
```

$$\Big(3\,(\mathrm{a13\,a22\,a31 - a12\,a23\,a31 - a13\,a21\,a32 + a11\,a23\,a32 + a12\,a21\,a33 - a11\,a22\,a33})^2\Big) \Big/ \Bigg(\Bigg(-\bigg(\Big(\mathrm{a12} - \frac{\mathrm{a13}}{x\,y^2}\Big)\Big(\mathrm{a11} - \frac{\mathrm{a13}}{x^2\,y}\Big) + \Big(\mathrm{a22} - \frac{\mathrm{a23}}{x\,y^2}\Big)\Big(\mathrm{a21} - \frac{\mathrm{a23}}{x^2\,y}\Big) + \Big(\mathrm{a32} - \frac{\mathrm{a33}}{x\,y^2}\Big)\Big(\mathrm{a31} - \frac{\mathrm{a33}}{x^2\,y}\Big)\bigg)^2 + \bigg(\Big(\mathrm{a12} - \frac{\mathrm{a13}}{x\,y^2}\Big)^2 + \Big(\mathrm{a22} - \frac{\mathrm{a23}}{x\,y^2}\Big)^2 + \Big(\mathrm{a32} - \frac{\mathrm{a33}}{x\,y^2}\Big)^2\bigg) \bigg(\Big(\mathrm{a11} - \frac{\mathrm{a13}}{x^2\,y}\Big)^2 + \Big(\mathrm{a21} - \frac{\mathrm{a23}}{x^2\,y}\Big)^2 + \Big(\mathrm{a31} - \frac{\mathrm{a33}}{x^2\,y}\Big)^2\bigg)\Bigg) \Big(\big(\mathrm{a13}\,(\mathrm{a21}\,x - \mathrm{a22}\,y) + \mathrm{a12}\,y\,(\mathrm{a23} - \mathrm{a21}\,x^2\,y) + \mathrm{a11}\,x\,(-\mathrm{a23} + \mathrm{a22}\,x\,y^2)\big)^2 + \big(\mathrm{a13}\,(\mathrm{a31}\,x - \mathrm{a32}\,y) + \mathrm{a12}\,y\,(\mathrm{a33} - \mathrm{a31}\,x^2\,y) + \mathrm{a11}\,x\,(-\mathrm{a33} + \mathrm{a32}\,x\,y^2)\big)^2 + \big(\mathrm{a23}\,(\mathrm{a31}\,x - \mathrm{a32}\,y) + \mathrm{a22}\,y\,(\mathrm{a33} - \mathrm{a31}\,x^2\,y) + \mathrm{a21}\,x\,(-\mathrm{a33} + \mathrm{a32}\,x\,y^2)\big)^2\Big)\Bigg)$$

We compute the distance to the tangent plane, so that we may compare its fourth power to the Gaussian curvature.

```
d = (u.r) / (Sqrt[u.u]) // Simplify;
```

```
d^4 // Simplify
```

```
(81 (a13 a22 a31 - a12 a23 a31 - a13 a21 a32 +
       a11 a23 a32 + a12 a21 a33 - a11 a22 a33)^4 x^4 y^4) /
 ((a13 (a21 x - a22 y) + a12 y (a23 - a21 x^2 y) + a11 x (-a23 + a22 x y^2))^2 +
    (a13 (a31 x - a32 y) + a12 y (a33 - a31 x^2 y) +
        a11 x (-a33 + a32 x y^2))^2 + (a23 (a31 x - a32 y) +
        a22 y (a33 - a31 x^2 y) + a21 x (-a33 + a32 x y^2))^2)^2
```

```
k / d^4 // Simplify
```

```
1 / (27 (a13 a22 a31 - a12 a23 a31 -
       a13 a21 a32 + a11 a23 a32 + a12 a21 a33 - a11 a22 a33)^2)
```

We now see that the transformed surface is a Tzitzeica surface with the same constant of proportionality (up to a volume term).

```
k Det[{{a11, a12, a13}, {a21, a22, a23}, {a31, a32, a33}}]^2 /
   d^4 // Simplify
```

$$\frac{1}{27}$$

4. The Role of Asymptotic Curves

In this section we present the connection between Tzitzeica curves and Tzitzeica surfaces. In particular, we show that asymptotic curves in a Tzitzeica surface must be Tzitzeica curves. Recall that a curve α in a surface M is asymptotic if its acceleration is always tangent to M:

$$\alpha'' \cdot U = 0. \tag{7}$$

In this case, the surface normal $\boldsymbol{U}$ and the Frenet binormal $\mathbf{B}$ coincide, so that the osculating plane of the curve and the surface tangent plane coincide at each point of a. It follows that

$$d_{\text{osc}} = d_{\text{tan}}. \tag{8}$$

Furthermore, it can be shown ([7], p.230) that the torsion and Gaussian curvature are related by

$$K = -\tau^2. \tag{9}$$

If we suppose that α is an asymptotic curve in a Tzitzeica surface M, then we have for some $a \in \mathbb{R}$,

$$K = a\, {d_{\tan}}^4. \tag{10}$$

Substituting for $d_{\tan}$ from (8) and for K from (9), we get

$$-\tau^2 = a\, {d_{\mathrm{osc}}}^4, \tag{11}$$

leading to the defining condition for α to be a Tzitzeica curve,

$$\tau = \tilde{a}\, {d_{\mathrm{osc}}}^2, \tag{12}$$

where $\tilde{a} = -a$.

From equation (9), it follows that a surface with asymptotic curves must have nonpositive curvature. In particular, the previously presented surface $z = 1/(x\,y)$ will not have asymptotic curves. To see an example of nontrivial asymptotic/Tzitzeica curves on a Tzitzeica surface, consider the surface $z = 1 \big/ \left(x^2 + y^2\right)$, which is indeed a Tzitzeica surface. It can be verified that the invariant ratio $\kappa \big/ d^4$ is $-4/27$.

Below is a standard plot of a portion of the surface in cylindrical coordinates. It is not too difficult to convince oneself using this plot and one's imagination that the fourth power of the distance from the origin to the tangent plane is proportional to the curvature at each point. For example, consider the region between the flat base and the narrow funnel parts, where the magnitude of the curvature is modestly positive. This is precisely where the tangent plane is nearly perpendicular to the position vector, and so there is a correspondingly modest distance from the origin to the tangent plane that is approximately equal to the length of the position vector. As one moves away from this region to a region of smaller curvature (down into the flatter base), the tangent plane is now nearly horizontal and nearly passes through the origin, indicating a correspondingly short distance between the origin and the tangent plane.

```
ParametricPlot3D[{ρ Cos[θ], ρ Sin[θ], 1/ρ^2}, {ρ, -4, 4},
  {θ, 0, Pi}]
```



Using standard methods to compute asymptotic curves (see for example Chapter 18 in [1]), we find the asymptotic curves to be the images of logarithmic spirals of the form

$$\theta = v \pm \sqrt{3}\ \log[\rho] \tag{13}$$

$$\rho[\theta] \to u\, e^{\frac{\pm\theta}{\sqrt{3}}}, \tag{14}$$

where u and v are new parameters. We plot one of the spirals to aid in visualization.

```
ParametricPlot[{Cos[θ] E^(θ/√3), Sin[θ] E^(θ/√3)}, {θ, -4, 4}]
```



By converting to the coordinates u and v, we obtain a parametrization whose constant coordinate curves are asymptotic/Tzitzeica curves on the surface. Here is one part of the surface with such a parametrization, making explicit the Tzitzeica curves.

$$\mathbf{ParametricPlot3D}\left[\left\{e^{\frac{v}{2\sqrt{3}}}\sqrt{u}\,\mathbf{Cos}\left[\frac{1}{2}\left(v-\sqrt{3}\,\mathbf{Log}[u]\right)\right],\ e^{\frac{v}{2\sqrt{3}}}\sqrt{u}\,\mathbf{Sin}\left[\frac{1}{2}\left(v-\sqrt{3}\,\mathbf{Log}[u]\right)\right],\ \frac{e^{-\frac{v}{\sqrt{3}}}}{u}\right\},\ \{u,\ .1,\ 5\},\ \{v,\ -5,\ 5\}\right]$$



5. Conclusion

In this article we have considered the Klein viewpoint of geometry in the case where the symmetry group is the affine group (or one of its subgroups). This results in the study of affine differential geometry. One of the earliest results in this direction was the aforementioned papers of Gheorghe Tzitzeica [3, 4], where he established the class of affine-invariant curves and surfaces that bear his name. Although the concept has been generalized to the case of hypersurfaces in arbitrary dimensions, it is the case of curves and surfaces in three dimensions that is particularly suitable for discussion at relatively elementary levels. Moreover, the subject is still an active area of research, though perhaps not as active as in the previous century, when the basic theory and results were being worked out. Lastly, we have given two detailed examples of how well-suited the material is, visually and computationally, for using *Mathematica*. There are many more activities and projects that can be performed with this material.

6. Appendix: The Tzitzeica Test

Here we summarize the commands to test if a given surface patch is Tzitzeica.

- Fill in the coordinates x, y, z of your surface as functions of the surface parameters $\boldsymbol{u}$ and $\boldsymbol{v}$.
- In order to effect a centro-affine transformation, multiply $\boldsymbol{r}$ by the desired nonsingular matrix (numerical or symbolic).

As is typical in these kinds of computations, each surface may require specific simplification commands at various steps. We find that for simple surfaces, the `Simplify` command usually suffices.

The Tzitzeica Test:

```
x =

y =

z =

r = {x, y, z}

ru = D[r, u]

rv = D[r, v]
```

```
e = ru.ru
```

```
f = ru.rv
```

```
g = rv.rv
```

```
normal = Cross[ru, rv] / Sqrt[Cross[ru, rv].Cross[ru, rv]] //
  Simplify
```

```
ruu = D[ru, u]
```

```
ruv = D[ru, v]
```

```
rvv = D[rv, v]
```

```
l = normal.ruu // Simplify
```

```
m = normal.ruv
```

```
n = normal.rvv // Simplify
```

```
k = (l n - m^2) / (e g - f^2) // Simplify
```

```
d = (normal.r) / (Sqrt[normal.normal]) // Simplify
```

```
d^4 // Simplify
```

```
k / d^4 // Simplify
```

References

[1] A. Gray, *Modern Differential Geometry of Curves and Surfaces with Mathematica*, 2nd ed., Boca Raton: CRC Press, 1998.

[2] F. Klein, "A Comprehensive Review of Recent Researches in Geometry," M. W. Haskell, trans., *Bulletin of the New York Mathematical Society,* **2**(10), 1893 pp. 215-249.

[3] G. Tzitzeica, "Sur une nouvelle classe de surfaces," *Les Comptes Rendus de l'Académie des sciences*, **144,** 1907 pp. 1257-1259.

[4] G. Tzitzeica, "Sur une nouvelle classe de surfaces," *Rendiconti del Circulo Matematico di Palermo*, **25**, 1908 pp. 180-187; **28**, 1909 pp. 210-216.

[5] W. Blaschke, *Vorlesungen über Differentialgeometrie II,* New York: Chelsea Publishing Company, 1967.

[6] K. Nomizu and T. Sasaki, *Affine Differential Geometry*, Cambridge, UK: Cambridge University Press, 1994.

[7] B. O'Neill, *Elementary Differential Geometry*, New York: Academic Press, 1966.

A. F. Agnew, A. Bobe, W. G. Boskoff, and B. D. Suceava, "Tzitzeica Curves and Surfaces," *The Mathematica Journal*, 2010. dx.doi.org/doi:10.3888/tmj.12-3.

About the Authors

Alfonso F. Agnew is a mathematical physicist interested particularly in geometric methods. He is an associate professor in the Mathematics Department at California State University at Fullerton.

Alexandru Bobe is an assistant in the Department of Mathematics at Ovidius University of Constanta. His main research interest is computational geometry.

Wladimir G. Boskoff is a professor in the Faculty of Mathematics and Computer Science at Ovidius University of Constanta, Romania. He is particularly interested in metric geometry and differential geometry.

Bogdan D. Suceava is an assistant professor in the Department of Mathematics at California State University at Fullerton. His main research interest is in submanifold geometry.

Alfonso F. Agnew
Department of Mathematics
California State University Fullerton
Fullerton, CA, 92834 USA
aagnew@fullerton.edu

Alexandru Bobe
Department of Mathematics and Computer Science
Ovidius University
Bd. Mamaia 124
Constanta, 900527 Romania
alexb@univ-ovidius.ro

Wladimir G. Boskoff
Department of Mathematics and Computer Science
Ovidius University
Bd. Mamaia 124
Constanta, 900527 Romania
boskoff@univ-ovidius.ro

Bogdan D. Suceava
Department of Mathematics
California State University Fullerton
Fullerton, CA, 92834 USA
bsuceava@fullerton.edu